# $Id$
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Simon Perry <aur [at] sanxion [dot] net>
# Contributor: Jonas Nyrén <jonas.nyren*mindkiller.com>

pkgname=libsidplayfp
pkgver=1.8.8
pkgrel=1
pkgdesc='Library to play Commodore 64 music derived from libsidplay2'
url='http://sourceforge.net/projects/sidplay-residfp/'
arch=('x86_64')
license=('GPL')
depends=('gcc-libs')
makedepends=('vice' 'xa')
optdepends=('vice: better SID support')
source=("http://downloads.sourceforge.net/project/sidplay-residfp/${pkgname}/${pkgver%.*}/${pkgname}-${pkgver}.tar.gz")
sha256sums=('d28b0c28ceea62627c137b4194e8bed866c8f2cec65bac97fde63639250a31e0')

build() {
	cd "${pkgname}-${pkgver}"
	./configure --prefix=/usr
	make
}

package() {
	cd "${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install

	install -d "${pkgdir}/usr/share/sidplayfp"
	ln -t "${pkgdir}/usr/share/sidplayfp" -s /usr/lib/vice/C64/{kernal,basic,chargen}
}
